/*
Ilene Hollin	
Proposal title: Price Disclosure Laws for Direct-to-Consumer Pharmaceutical Advertising: Price Transparency, Information Asymmetry and Consumer Behavior


HYPOTHESES

Stated-Hyp1: Price disclosure influences information asymmetry, intention to comparison shop, and intent to seek treatment (p. 2).

	Test-Hyp1: The level of information assymtery will vary between single list price vs. control conditon.
	
	Test-Hyp2: The level of information assymtery will vary between single list price + avg consumer cost vs. control.	

Stated-Hyp2: Price disclosure influences consumer behavior in terms of intention to comparison shop and intention to seek treatment (p. 2).

	Test-Hyp3: The intent to comparison shop will vary between single list price vs. control conditon. 
	
	Test-Hyp4: The intent to comparison shop will vary between single list price + avg consumer cost vs. control.	

	Test-Hyp5: The intent to seek treatment will vary between single list price vs. control conditon. 
	
	Test-Hyp6: The intent to seek treatment will vary between single list price + avg consumer cost vs. control.		

Stated-Hyp3: Disclosure of list price influences consumer behavior in terms of intention to comparison shop and intention to seek treatment, and this effect is moderated by condition - i.e. whether condition is life threatening and whether condition is symptomatic (p. 3)

	Test-Hyp7: Life threatening condition vs. non-life threatening condition will moderate the effect of price disclosure on intent to comparison shop.
	Test-Hyp8: Symptomatic condition vs. non-symptomatic condition will moderate the effect of price disclosure on intent to comparison shop.
	
	Test-Hyp9: Life threatening condition vs. non-life threatening condition will moderate the effect of price disclosure on intent to seek treatment.
	Test-Hyp10: Symptomatic condition vs. non-symptomatic condition will moderate the effect of price disclosure on intent to seek treatment.	

Stated-Hyp4: Certain factors influence the percieved afforadability, intention to seek treatment, and intention to comparison shop (p. 3, research question 3)

	// NOTE: there is no formal hypothesis to be tested.

Stated-Hyp5: Affordability perception mediates the effect of list price disclosure on intention to seek treatment and intention to comparison shop.	(conditional on these intentions being affected by price disclosure) p. 3
	
	Test-Hyp11: Affordability will mediate the effect of price disclosure on intent to comparison shop.
	Test-Hyp12: Affordability will mediate the effect of price disclosure on intent to seek treatment.
	
********************************************************************************
NOTES
- there are many differences between the proposed questions in the appendix an 
the questions that were actually fielded. We have used the questions that most 
closely matched the intended questions from the proposal. 
	
- the proposal just talks about one treatment (price disclosure), but the 
appendix introduces two variations of this treatment: list price and out of 
pocket. In terms of construct underlying the treatment (i.e. price disclosure) 
both treatments are variations of each other. we treat this a single treatment 
(i.e. whether or not price was disclosed).

- For information asymmetry, we are using the measure labeled as the IA measure 
in the proposal.

- The proposal said they will test mediation only if the primary relationships 
are significant. Because there is no effect of price disclosure on intentions, 
mediation analyses were not conducted. 

- The authors mention using ANCOVA in the analysis of stated hypothesis 3. 
We are using OLS for standardization.
*/

clear all
use "Hollin1121.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS

	tab P_CONDITION

	* symptomatic or not
	recode P_CONDITION (2/3=1) (1=0), gen(symptomatic)
	tab symptomatic	
	
	* life threatening or not
	recode P_CONDITION (1/2=1) (3=0), gen(lifethreatening)
	tab lifethreatening	
	
	* price transparency
	tab P_TRANSP
	
* OUTCOME MEASURES

* information asymmetry
		/*
		The ad provided enough information for me to decide if I should discuss the medication with a doctor.
		Note: there are two questions author wanted to include as measure of this construct, but only one was fielded. See page 18 of proposal.*/
 	tab Q1
	replace Q1=. if Q1>7
	gen information=7-Q1
	tab information // higher values indicate greater intent	
	
* intent to seek treatment
	*How likely are you to ask your doctor for a prescription for the advertised medication?
	tab Q4	
	replace Q4=. if Q4>7
	gen intent_seektreat=7-Q4
	tab intent_seektreat // higher values indicate greater intent
	
* intent to comparison shop
	
	/*Having seen this ad, how likely are you to compare this medication with other prescription medications based on price?	
	note: intended questions from the quex were changed before fielding. In the fielded quex, there is another question about comparison shopping but it's not about price. I'm using the question about price only.
	*/
	
	tab Q5	
	replace Q5=. if Q5>7
	gen intent_compshop=7-Q5
	tab intent_compshop // higher values indicate greater intent
	
* MEDIATOR

* affordability
	tab Q8C
	replace Q8C=. if Q8C>5
	gen affordability = 6-Q8C
	tab affordability // higher values indicate greater affordability
	
********************************************************************************

* ANALYSIS

*Test-Hyp1: The level of information assymtery will vary between single list price vs. control conditon.
	reg information i.P_TRANSP if P_TRANSP!=3
	//do not reject. 0.006
	tess 2.P_TRANSP, init(Hollin1121) bonf(3)
	
*Test-Hyp2: The level of information assymtery will vary between single list price + avg consumer cost vs. control.		
	reg information i.P_TRANSP if P_TRANSP!=2
	tess 3.P_TRANSP, bonf(3)
	
	
*Test-Hyp3: The intent to comparison shop will vary between single list price vs. control conditon. 
	reg intent_compshop i.P_TRANSP if P_TRANSP!=3
	// reject. 
	tess 2.P_TRANSP, bonf(3)
	
*Test-Hyp4: The intent to comparison shop will vary between single list price + avg consumer cost vs. control.	
	reg intent_compshop i.P_TRANSP if P_TRANSP!=2
	// reject. 
	tess 3.P_TRANSP, bonf(3)
	
*Test-Hyp5: The intent to seek treatment will vary between single list price vs. control conditon. 
	reg intent_seektreat i.P_TRANSP if P_TRANSP!=3
	// reject. 
	tess 2.P_TRANSP, bonf(3)
	
*Test-Hyp6: The intent to seek treatment will vary between single list price + avg consumer cost vs. control.			
	reg intent_seektreat i.P_TRANSP if P_TRANSP!=2
	// reject. 
	tess 3.P_TRANSP, bonf(3)	

		
	
*Test-Hyp7: Life threatening condition vs. non-life threatening condition will moderate the effect of price disclosure on intent to comparison shop.
	reg intent_compshop i.P_TRANSP##i.lifethreatening if P_TRANSP!=3
	// reject. 0.259
	tess 2.P_TRANSP#1.lifethreatening, bonf(2)
	
*Test-Hyp8: Symptomatic condition vs. non-symptomatic condition will moderate the effect of price disclosure on intent to comparison shop.
	reg intent_compshop i.P_TRANSP##i.symptomatic if P_TRANSP!=3
	// reject. 0.745
	tess 2.P_TRANSP#1.symptomatic, bonf(2)	
		
*Test-Hyp9: Life threatening condition vs. non-life threatening condition will moderate the effect of price disclosure on intent to seek treatment.
	reg intent_seektreat i.P_TRANSP##i.lifethreatening if P_TRANSP!=2
	// reject. 0.136
	tess 3.P_TRANSP#1.lifethreatening, bonf(2)

*Test-Hyp10: Symptomatic condition vs. non-symptomatic condition will moderate the effect of price disclosure on intent to seek treatment.	
	reg intent_seektreat i.P_TRANSP##i.symptomatic if P_TRANSP!=2
	// reject. 0.696
	tess 3.P_TRANSP#1.symptomatic, bonf(2)

